N71-29922

NASA TECHNICAL MEMORANDUM

NASA TM X-2320

NASA TM X-2320

COMPUTER PROGRAM FOR EVALUATION OF BLOCH-GRUENEISEN PARAMETERS OF METALS AND EVALUATION OF ELECTRICAL RESISTIVITY OF TANTALUM AS A FUNCTION OF TEMPERATURE

by Thor T. Semler and John P. Riehl

Lewis Research Center
Cleveland, Ohio 44135

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • JULY 1971

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA TM X-2320		

4. Title and Subtitle	5. Report Date
COMPUTER PROGRAM FOR EVALUATION OF BLOCH-GRUENEISEN PARAMETERS OF METALS AND EVALUATION OF ELECTRICAL RESISTIVITY OF TANTALUM AS A FUNCTION OF TEMPERATURE	July 1971
	6. Performing Organization Code
7. Author(s)	8. Performing Organization Report No.
Thor T. Semler and John P. Riehl	E-6242
	10. Work Unit No.
9. Performing Organization Name and Address	129-02
Lewis Research Center National Aeronautics and Space Administration Cleveland, Ohio 44135	11. Contract or Grant No.
	13. Type of Report and Period Covered
12. Sponsoring Agency Name and Address	Technical Memorandum
National Aeronautics and Space Administration Washington, D.C. 20546	14. Sponsoring Agency Code

15. Supplementary Notes

16. Abstract

A computer program has been written for the least squares evaluation of the parameters in the Bloch-Grueneisen relation for "ideal" electrical resistivity of metals. The program input is experimental measurements of resistivity and temperature. The constants associated with tantalum have been determined by using this code and several experimental measurements of the electrical resistivity of tantalum.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Tantalum; Electrical resistivity; Bloch; Grueneisen; Temperature; Resistance; Davidon; Nonlinear; Least squares	Unclassified - unlimited

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified	27	$3.00

*For sale by the National Technical Information Service, Springfield, Virginia 22151

COMPUTER PROGRAM FOR EVALUATION OF BLOCH-GRUENEISEN PARAMETERS OF METALS AND EVALUATION OF ELECTRICAL RESISTIVITY OF TANTALUM AS A FUNCTION OF TEMPERATURE

by Thor T. Semler and John P. Riehl

Lewis Research Center

SUMMARY

A computer program using nonlinear functional minimization has been written to obtain least squares solutions, from experimental measurements, for the constant terms of the Bloch-Grueneisen relation

$$\rho_T(T) = A\left(\frac{T}{\theta_R}\right)^5 \int_0^{\theta_R/T} \frac{x^5\,dx}{(e^x - 1)(1 - e^{-x})}$$

where $\rho_T(T)$ is the "ideal" electrical resistivity of metals as a function of temperature, A is a constant of the metal, T is the temperature in K, and θ_R is the characteristic temperature of resistivity.

The metal tantalum has been analyzed by using the code, and a typical result is $\theta_R = 217.54$ K and $A = 39.95$ microohm-centimeters between 10 and 250 K.

INTRODUCTION

A fundamental constant in Ohm's law is the electric resistance of the conductor R. For a particular conductor of length l (in cm) and uniform cross-sectional area a (in cm^2), R (in ohms) may be computed as

$$R = \frac{\rho l}{a} \tag{1}$$

where ρ is the electrical resistivity in ohm-centimeters. As the electrical resistance is a quantity of great interest in both engineering and solid-state physics (refs. 1 to 3), it is important to be able to determine the electrical resistivity and then compute the resistance of a conductor.

The resistivity of a metal is a function of temperature. On approaching very low temperatures, near absolute zero, the electrical resistivity assumes a constant value (neglecting the region in which some metals become superconductors) ρ_0, called the residual resistivity. This quantity ρ_0 arises from imperfections, impurities, and strains in the metal lattice and must be determined for each individual sample.

The total resistivity ρ may be divided into two portions, the residual resistivity and the temperature-dependent resistivity $\rho_T(T)$:

$$\rho = \rho_0 + \rho_T(T) \tag{2}$$

This division is known as Matthiessen's rule (refs. 3 and 4).

It is possible to derive a formula for the temperature-dependent resistivity $\rho_T(T)$ over a large temperature range from certain approximations about the interactions of conduction electrons and the metallic lattice vibrations (refs. 5 to 8). The formula (3) so derived is referred to as the Bloch-Grueneisen relation

$$\rho_T(T) = A\left(\frac{T}{\theta_R}\right)^5 \int_0^{\theta_R/T} \frac{x^5\,dx}{(e^x - 1)(1 - e^{-x})} \tag{3}$$

where A is a constant of the metal and θ_R is the characteristic temperature of resistivity. The Bloch-Grueneisen relation is widely applied because it provides a good approximation for the temperature-dependent resistivity for many metals.

Because of the difficulties in the form of the relation, many rule-of-thumb techniques have been evolved by experimenters to evaluate A and θ_R from experimental data. Unfortunately, many of these rule-of-thumb techniques are rather crude (ref. 4).

The computer program described in the section ANALYSIS allows one to compute in the least-squares sense the best values of A and θ_R from all the experimental data one might wish to use. This computer program is then used to obtain values of A and θ_R for metallic tantalum.

ANALYSIS

Given values of ρ_T that have been measured as a function of temperature, one would like the values of θ_R and A which are "best" in the least-squares sense. This means that one must form the function $f(A,\theta) = \sum_i \left[\rho_{T,i,\text{measured}} - \rho_{T,i,\text{calculated}}\right]^2$ and minimize it by the variation of θ_R and A.

By the rules of ordinary calculus, $f(A,\theta)$ without constraint obtains a local minimum or reaches a saddle point when the gradient $\nabla f(A,\theta) = 0$ at particular values of A and θ. For those functions $l(x,p)$ that are linear in p the gradient requirement produces a set of simultaneous linear equations. But for functions that are nonlinear in p, this requirement is not easily met. One is confronted with a set of nonlinear simultaneous equations in p to be solved. Such is the case for the Bloch-Grueneisen equation. Its gradient has components

$$\frac{\partial f}{\partial A} = -2 \sum_{i=1}^{NDP} \left\{ \left[\rho_T(T_i) - A\left(\frac{T_i}{\theta_R}\right)^5 \int_0^{\theta_R/T_i} \frac{x^5\,dx}{(e^x - 1)(1 - e^{-x})} \right] \cdot \left(\frac{T_i}{\theta_R}\right)^5 \int_0^{\theta_R/T_i} \frac{x^5\,dx}{(e^x - 1)(1 - e^{-x})} \right\}$$

$$\frac{\partial f}{\partial \theta} = -2 \sum_{i=1}^{NDP} \left\{ \left[\rho_T(T_i) - A\left(\frac{T_i}{\theta_R}\right)^5 \int_0^{\theta_R/T_i} \frac{x^5\,dx}{(e^x - 1)(1 - e^{-x})} \right] \cdot \left(-5A \frac{T_i^5}{\theta_R^6} \int_0^{\theta_R/T_i} \frac{x^5\,dx}{(e^x - 1)(1 - e^{-x})} + \frac{A}{T_i} \frac{1}{\left(e^{\theta_R/T_i} - 1\right)\left(1 - e^{-\theta_R/T_i}\right)} \right) \right\}$$

where NDP is the number of data points.

The second component was arrived at by using the chain rule and Leibnitz's rule, which is, if $F(x) = \int_{\varphi(x)}^{\psi(x)} f(x,y)dy$ is continuously differentiable, then

$$F'(x) = \int_{\varphi(x)}^{\psi(x)} \frac{\partial}{\partial x} f(x,y)dy - f(x, \varphi(x)) \cdot \varphi'(x) + f(x, \psi(x)) \cdot \varphi'(x)$$

To accomplish the unconstrained function minimization, the method of Fletcher and Powell was employed (ref. 9). The Fletcher and Powell algorithm is a modification of a method of Davidon (ref. 10). It is a powerful and general method for finding the local minimum of a general function $f(x)$.

Central to the method is a symmetric positive definite matrix $\mathscr{H}_i$ which is updated at each iteration i. The current direction of motion $\vec{S}_i$ is supplied by $\mathscr{H}_i$ when it is multiplied with the current change gradient vector. An iteration is described by the following: If $\mathscr{H}_0$ is any positive definite matrix, usually the identity matrix $\mathscr{I}$, on the first iteration only, then

$$\vec{S}_i = -\mathscr{H}_i \nabla f(\vec{x})\big|_{\vec{x}=\vec{x}_i}$$

Choose $\alpha = \alpha_i$ by minimizing $f(\vec{x}_i + \alpha\vec{S}_i)$; this straight line minimization is done with cubic interpolation:

$$\vec{\sigma}_i = \alpha\vec{S}_i$$

$$\vec{x}_{i+1} = \vec{x}_i + \vec{\sigma}_i$$

$$\mathscr{H}_{i+1} = \mathscr{H}_i + \mathscr{A}_i + \mathscr{B}_i$$

where the matrices $\mathscr{A}_i$ and $\mathscr{B}_i$ are defined by

$$\mathscr{A}_i = \frac{\vec{\sigma}_i(\widetilde{\sigma}_i)}{(\overline{\sigma}_i)\vec{y}_i}, \quad \vec{y}_i = \nabla f(\vec{x}_{i+1}) - \nabla f(\vec{x}_i)$$

and

$$\mathscr{B}_i = \frac{-\mathscr{H}_i \vec{y}_i \widetilde{y}_i \mathscr{H}_i}{\widetilde{y}_i \mathscr{H}_i \vec{y}_i}$$

$\widetilde{y}_i$ being the transpose of $\vec{y}_i$.

The numerators of $\mathscr{A}_i$ and $\mathscr{B}_i$ are both matrices, while the denominators are scalars. Fletcher and Powell (ref. 9) prove the following:

(1) The matrix $\mathscr{H}_i$ is positive and definite for all i. As a consequence, the method will always converge since

$$\frac{d}{d\alpha} f(\vec{x}_i + \alpha\vec{S}_i) \Big|_{\alpha=0} = -\widetilde{\nabla f(\vec{x}_i)}\, \mathscr{H}_i\, \nabla f(\vec{x}_i) < 0$$

That is, the function f is initially decreasing along the direction $\vec{S}_i$. So that the function can be decreased at each iteration by minimizing along $\vec{S}_i$.

(2) When the method is applied to the quadratic matrix equation $q(\vec{x}) = a + \widetilde{b}\vec{x} + \widetilde{x}\mathscr{A}\overline{x}$ and $\vec{x}$ is a vector of n dimensions,

(a) The directions $\vec{S}_i$ (or equivalently $\vec{\sigma}_i$) are $\mathscr{A}$ conjugate, that is, $\widetilde{S}_i\ \vec{S}_j = 0$ for $i \neq j$. This condition leads to a minimum in n steps.

(b) The matrix $\mathscr{H}_i$ converges to the inverse of the Hessian, that is, the matrix of second partial derivatives after n iterations, $\mathscr{H}_n = \mathscr{A}^{-1}$. When applied to a general function $f(\vec{x})$, $\mathscr{H}_i$ tends to the inverse of the Hessian evaluated at the minimum.

The Fletcher-Powell algorithm is represented by the flow chart in figure 1.

CALCULATION OF BLOCH-GRUENEISEN RELATION

The integral portion of the Bloch-Grueneisen relation was calculated by using a modified Simpson's rule integration scheme. This scheme, programmed as subprogram SIMPS1, adapts to regions where more points are required to obtain an accurate result. Had the integral been too expensive (in terms of computer time) to compute, a spline approximation to tabular results of the integral could have been made. It was our experience that the Bloch-Grueneisen relation and the ensuing least-squares function could be calculated in very little time by using the modified Simpson's rule routine.

In the program the exponent 5 of equation (3) may be varied, as indicated in the comments card. This permits the user to use other than a fifth-order Bloch-Grueneisen relation. Input is described in appendix A. A flow chart of the main program is shown in figure 2. A listing of the program is given in appendix B.

DATA USED IN EVALUATION OF TANTALUM

Experimental measurements of the electrical resistivity of the metal tantalum have been analyzed by using this code. Only experimental values of the electrical resistivity in the temperature region from above 0 K (actually 10 K as tantalum is a superconductor) to about 400 K have been used in the program. The values have been taken from this region since the least-squares fit of the parameters is relatively insensitive to data outside the range 0 K to a few times θ_R K.

Cox Data in Temperature Range 77 to 373 K

Cox in 1943 performed a series of experiments to determine both the thermal and electrical conductivities of tungsten and tantalum (ref. 11). This series of experiments yielded three values of the electrical resistivity of tantalum.

The sample used was a tantalum wire about 40 centimeters long and about 0.0254 centimeter in diameter. The wire was aged by passing as high a current as possible through it without evaporating it. The tantalum sample was aged at both 1800° and 2000° C for a total time of 2750 hours. The resistance at zero power input was plotted at a function of aging time. The resistance decreased rapidly at first and finally reached a constant value; at that point, aging was ceased. The chemical purity quoted for the sample was 99.9 percent.

After aging, the sample was immersed in baths of boiling liquid nitrogen, ice water, and boiling water, readings of voltage and current across the sample were taken, and the resistivity was computed; these data are given in table I.

White and Woods Data in Temperature Range 10 to 295 K

White and Woods, in a series of experiments to determine the electrical and thermal resistivity of the transition elements, report 21 values of the ideal resistivity of tantalum (ref. 12). These results were obtained by subtracting the residual resistivity from the total resistivity of the sample at a temperature, and they are shown in table II.

The specimen was mounted in a cryostat. One end of the specimen was soldered to a copper post, and a specimen heater was attached to the other end. Copper wires were attached to intermediate points of the rod to act as electrical potential leads for the resistivity measurements. The specimen was surrounded by low-pressure helium gas to preserve temperature equilibrium in the cryostat.

The purities of the three samples of tantalum used are quoted as 99.9 percent, "high", and 99.9 percent. All samples were vacuum annealed to remove as much work hardening as possible.

White and Woods suggest that the electrical resistivity of tantalum at lower temperatures follows more nearly a $T^{3.8}$ proportionality than a T^5 proportionality. However, in the section of their report devoted to the error analysis of the electrical resistivity measurements they indicate the difficulties in determining the low-temperature ideal resistivities.

RESULTS AND DISCUSSION

The program was executed by using 19 of White and Woods experimental values. This involved a temperature range from 10 to 250 K. The resultant values of the parameters were A = 39.95 microohm-centimeters and $\theta_R = 217.54$ K. The resultant fit of the calculated data to the experimental points is shown in figure 3. It can be seen that the agreement is excellent. The tabular results are shown in appendix C.

The three values of Cox were analyzed by using the code. This involved a temperature range of 77.3 to 373.4 K. The result of this analysis were A = 39.51 microohm-centimeters and $\theta_R = 210.77$ K. The resulting fit of the data is in figure 4.

Because of the rather limited nature of the Cox data, both the White and Woods data and the Cox data were analyzed together. The resulting fit is shown in figure 5. It can be clearly seen that the Cox data appear higher than the White and Woods data. This indicates either that the residual resistivity of the Cox sample had not been subtracted from the individual values or that the sample had been insufficiently annealed. Thus, the Cox data have been rejected in following the analysis.

As an illustrative example of the utility of the code, the White and Woods data have been analyzed parametrically by using the highest temperature involved as a parameter. The lowest temperature in all these cases was 10 K. The results are shown in figures 6 and 7. The results are tabulated in table III. They show both A and θ_R increasing to their asymptotic values. These are typical results for metallic samples (ref. 4). The characteristic temperature of resistivity θ_R is not to be confused with θ_D, the Debye characteristic temperature (ref. 13).

It should be indicated at this point that while these results might be obtained by extensive hand calculation, the results shown were obtained in a fraction of a minute by an IBM 7094-II computer.

SUMMARY OF RESULTS

A computer program was written for evaluation of Bloch-Grueneisen parameters of metals and evaluation of electrical resistivity of tantalum as a function of temperature, and the following results were obtained:

1. The values of the Bloch-Grueneisen parameters for the data of White and Woods were a constant of the metal $A = 39.95$ microohm-centimeters and a characteristic temperature of resistivity $\theta_R = 217.54$ K over the temperature range from 10 to 250 K.

2. The results of Cox apparently were not adjusted to ideal resistivity values.

3. When the Bloch-Grueneisen parameters were plotted as a function of the highest temperature used, for tantalum, they reached their asymptotic values at roughly a temperature of θ_R.

Lewis Research Center,
National Aeronautics and Space Administration,
Cleveland, Ohio, April 19, 1971,
129-02.

APPENDIX A

PROGRAM DESCRIPTION

Program Input Data

The card input to the program consists of a temperature T_i, $\rho_{T,i}$, the source of the data, and the first guess for A and θ_R. The format for these cards is 2F10.0,3X,A6. The user supplies as many cards with T_i, $\rho_{T,i}$, and the source as needed. The last card contains the first guess for A and θ_R with the field for the source left blank. This card causes the start of the least-squares curve fit. If the user wants the preceeding values of A and θ_R as first guesses, the last card should be left entirely blank.

The user may add coding of his own, starting after statement number 7 (card number 37). By punching END on a data card starting in card column 24, the user may execute this coding. This END card causes the program to transfer to statement 7.

Program Output

The program output consists of a tabulation of the input data along with $\rho_c T_i$, the calculated value of $\rho_{T,i}$ as defined by the Bloch-Grueneisen relation, and the difference between $\rho_c T_i$ and $\rho_{T,i}$. The values of A, θ_R, and the sum of the differences squared are also printed. An example of this output is found in appendix C.

APPENDIX B

FORTRAN LISTING OF PROGRAM AND OUTPUT

```
      COMMON /BLOCK/ T(100),RHO(100),RHOC(100),I
      COMMON /SPACE/ WORK(10)
      COMMON /ESTIM/ EST,EPS,LIMIT,IER
      EXTERNAL FAT
      DIMENSION GRAD(2), X(2), SOURCE(100)
      EQUIVALENCE (X(1),A), (X(2),THETA)
      DATA BLANK/1H /
      DATA END/3HEND/
      DATA N/2/
      WRITE (6,8)
1     I=1
2     READ (5,12) T(I),RHO(I),SOURCE(I)
      IF (SOURCE(I)-BLANK) 3,4,3
3     CONTINUE
      IF (SOURCE(I).EQ.END) GO TO 7
      I=I+1
      GO TO 2
4     CONTINUE
      IF (T(I).EQ.0.) GO TO 5
C     MAKE A FIRST GUESS AT A AND THETA
      A=T(I)
      THETA=RHO(I)
5     CONTINUE
      I=I-1
C     CALL THE FLETCHER - POWELL SUBROUTINE
      CALL FLTPWL (FAT,N,X,VAL,GRAD)
      IF (IER.NE.0) WRITE (6,9) IER
      WRITE (6,11)
      DO 6 J=1,I
      DIF=RHO(J)-RHOC(J)
      WRITE (6,13) SOURCE(J),T(J),RHO(J),RHOC(J),DIF
6     CONTINUE
      WRITE (6,10)
      WRITE (6,14) A,THETA,VAL
      WRITE (6,8)
      GO TO 1
7     CONTINUE
C     PERFORM ANY OTHER CALCULATIONS HERE
      STOP
C
C
8     FORMAT (1H1)
9     FORMAT (10X,4HIER=,I2)
10    FORMAT (1HK,24X,1HA,16X,5HTHETA,8X,23HSUM OF (DIFFERENCES)**2)
11    FORMAT (1HK,16X,6HSOURCE,17X,11HTEMPERATURE,13X,3HRHO,12X,14HRHO C
     1ALCULATED,8X,10HDIFFERENCE)
12    FORMAT (2F10.0,3X,A6)
13    FORMAT (17X,A6,7X,4F20.6)
14    FORMAT (10X,3F20.6)
      END
```

```
      SUBROUTINE FAT (N,Q,VLL,GRDD)
C     THIS SUBROUTINE CALCULATES THE LEAST SQUARES FUNCTION F(X)
C     AND THE GRADIENT  OF THE SAME
C     IN THIS PART X(1)=A , X(2)=THETA
      COMMON /BLOCK/ T(100),RHO(100),BGR(100),NCASES
      COMMON /EXPK/ K
      DIMENSION X(2), GRDD(1), Q(1), GRAD(2)
      EXTERNAL FUNKY
      DO 1 I=1,N
1     X(I)=Q(I)
      JOKE=0
      GO TO 2
      ENTRY BLAST(Z,Y,VLL)
      X(1)=Z
      X(2)=Y
      JOKE=1
2     CONTINUE
      VAL=0.
      DO 3 I=1,2
3     GRAD(I)=0.
      AK=K
      DO 4 I=1,NCASES
C     CALCULATE THE BLOCH - GRUENEISEN RELATIONSHIP
      X2DTI=X(2)/T(I)
      TIDX2=(T(I)/X(2))**K
      XX=SIMPS1(0.,X2DTI,FUNKY,L)
      BGR(I)=X(1)*TIDX2*XX
C     COMPUTE THE DIFFERENCE BETWEEN THE DATA AND THE B.G.R.
      DIF=RHO(I)-BGR(I)
      VAL=VAL+DIF**2
C     TRANSFER AROUND THE UNWANTED GRADIENT CALCULATIONS WHEN JOKE IS 1
      IF (JOKE.EQ.1) GO TO 4
      DIF=2.*DIF
C     CALCULATE THE COMPONENTS OF THE GRADIENT  (GRAD(1) AND GRAD(2))
      GRAD(1)=GRAD(1)-DIF*TIDX2*XX
      EXP1=EXP(X2DTI)
      EXPS=(EXP1+1./EXP1-2.)*T(I)
      GRAD(2)=GRAD(2)+(AK*BGR(I)/X(2)-X(1)/EXPS)*DIF
4     CONTINUE
      VLL=VAL
      IF (JOKE.EQ.1) RETURN
      DO 5 II=1,2
5     GRDD(II)=GRAD(II)
      RETURN
      END
```

```
      FUNCTION FUNKY (X)
C     THIS FUNCTION CALCULATES THE INTEGRAND IN THE B.G.R.
      COMMON /EXPK/ K
      IF (X.EQ.0.) GO TO 1
      EXP1=EXP(X)
      FUNKY=EXP1+1./EXP1-2.
      FUNKY=X**K/FUNKY
      RETURN
1     FUNKY=0.
      RETURN
      END
```

```
      BLOCK DATA
      COMMON /ESTIM/ EST,EPS,LIMIT,IER
      COMMON /EXPK/ K
C     THE ORDER OF THE BLOCH-GRUENEISEN RELATIONSHIP MAY BE CHANGED.
C     TO DO SO,CHANGE THE VALUE OF K IN THE FOLLOWING DATA STATEMENT.
      DATA K/5/
      DATA EST,EPS,LIMIT/1.E-2,1.E-5,1000/
      END
```

```
C
C     SUBROUTINE FLTPWL
C
C        PURPOSE
C           TO FIND A LOCAL MINIMUM OF A FUNCTION OF SEVERAL VARIABLES
C           BY THE METHOD OF FLETCHER AND POWELL
C
C     USAGE
C           CALL FLTPWL(FUNCT,N,X,F,G)
C
C        DESCRIPTION OF PARAMETERS
C           FUNCT  - USER-WRITTEN SUBROUTINE CONCERNING THE FUNCTION TO
C                    BE MINIMIZED. IT MUST BE OF THE FORM
C                    SUBROUTINE FUNCT(N,ARG,VAL,GRAD)
C                    AND MUST SERVE THE FOLLOWING PURPOSE
C                    FOR EACH N-DIMENSIONAL ARGUMENT VECTOR  ARG,
C                    FUNCTION VALUE AND GRADIENT VECTOR MUST BE COMPUTED
C                    AND, ON RETURN, STORED IN VAL AND GRAD RESPECTIVELY
C           N      - NUMBER OF VARIABLES
C           X      - VECTOR OF DIMENSION N CONTAINING THE INITIAL
C                    ARGUMENT WHERE THE ITERATION STARTS. ON RETURN,
C                    X HOLDS THE ARGUMENT CORRESPONDING TO THE
C                    COMPUTED MINIMUM FUNCTION VALUE
C           F      - SINGLE VARIABLE CONTAINING THE MINIMUM FUNCTION
C                    VALUE ON RETURN, I.E. F=F(X).
C           G      - VECTOR OF DIMENSION N CONTAINING THE GRADIENT
C                    VECTOR CORRESPONDING TO THE MINIMJM ON RETURN,
C                    I.E. G=G(X).
C           EST    - IS AN ESTIMATE OF THE MINIMUM FUNCTION VALUE.
C           EPS    - TESTVALUE REPRESENTING THE EXPECTED ABSOLUTE ERROR.
C                    A REASONABLE CHOICE IS 10**(-6), I.E.
C                    SOMEWHAT GREATER THAN 10**(-D), WHERE D IS THE
C                    NUMBER OF SIGNIFICANT DIGITS IN FLOATING POINT
C                    REPRESENTATION.
C           LIMIT  - MAXIMUM NUMBER OF ITERATIONS.
C           IER    - ERROR PARAMETER
C                    IER = 0 MEANS CONVERGENCE WAS OBTAINED
C                    IER = 1 MEANS NO CONVERGENCE IN LIMIT ITERATIONS
C                    IER =-1 MEANS ERRORS IN GRADIENT CALCULATION
C                    IER = 2 MEANS LINEAR SEARCH TECHNIQUE INDICATES
C                    IT IS LIKELY THAT THERE EXISTS NO MINIMUM.
C           H      - WORKING STORAGE OF DIMENSION N*(N+7)/2.
C
C        REMARKS
C            1) THE SUBROUTINE NAME REPLACING THE DUMMY ARGUMENT  FUNCT
C               MUST BE DECLARED AS EXTERNAL IN THE CALLING PROGRAM.
```

```
C          II) IER IS SET TO 2 IF , STEPPING IN ONE OF THE COMPUTED
C              DIRECTIONS, THE FUNCTION WILL NEVER INCREASE WITHIN
C              A TOLERABLE RANGE OF ARGUMENT.
C              IER = 2 MAY OCCUR ALSO IF THE INTERVAL WHERE F
C              INCREASES IS SMALL AND THE INITIAL ARGUMENT WAS
C              RELATIVELY FAR AWAY FROM THE MINIMUM SUCH THAT THE
C              MINIMUM WAS OVERLEAPED. THIS IS DUE TO THE SEARCH
C              TECHNIQUE WHICH DOUBLES THE STEPSIZE UNTIL A POINT
C              IS FOUND WHERE THE FUNCTION INCREASES.
C
C       SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C          FUNCT
C
C       METHOD
C          THE METHOD IS DESCRIBED IN THE FOLLOWING ARTICLE
C          R. FLETCHER AND M.J.D. POWELL, A RAPID DESCENT METHOD FOR
C          MINIMIZATION,
C          COMPUTER JOURNAL VOL.6, ISS. 2, 1963, PP.163-168.
C     THIS SUBROUTINE IS A MODIFICATION OF THE FMFP PROGRAM FROM THE IBM
C     SCIENTIFIC SUBROUTINE PACKAGE

C
      SUBROUTINE FLTPWL (FUNCT,N,X,F,G)
      COMMON /ESTIM/ EST,EPS,LIMIT,IER
      COMMON /SPACE/ H(1)
C
C        DIMENSIONED DUMMY VARIABLES
      DIMENSION X(1), G(1)
C
C        COMPUTE FUNCTION VALUE AND GRADIENT VECTOR FOR INITIAL ARGUMENT
      CALL FUNCT (N,X,F,G)
C
C        RESET ITERATION COUNTER AND GENERATE IDENTITY MATRIX
      IER=0
      KOUNT=0
      N2=N+N
      N3=N2+N
      N31=N3+1
    1 K=N31
      DO 4 J=1,N
      H(K)=1.
      NJ=N-J
      IF (NJ) 5,5,2
    2 DO 3 L=1,NJ
      KL=K+L
    3 H(KL)=0.
    4 K=KL+1
C
C        START ITERATION LOOP
    5 KOUNT=KOUNT+1
C
C        SAVE FUNCTION VALUE, ARGUMENT VECTOR AND GRADIENT VECTOR
      OLDF=F
      DO 9 J=1,N
      K=N+J
      H(K)=G(J)
      K=K+N
      H(K)=X(J)
C
C        DETERMINE DIRECTION VECTOR H
      K=J+N3
      T=0.
```

```
      DO 8 L=1,N
      T=T-G(L)*H(K)
      IF (L-J) 6,7,7
6     K=K+N-L
      GO TO 8
7     K=K+1
8     CONTINUE
9     H(J)=T
C
C        CHECK WHETHER FUNCTION WILL DECREASE STEPPING ALONG H.
      DY=0.
      HNRM=0.
      GNRM=0.
C
C        CALCULATE DIRECTIONAL DERIVATIVE AND TESTVALUES FOR DIRECTION
C        VECTOR H AND GRADIENT VECTOR G.
      DO 10 J=1,N
      HNRM=HNRM+ABS(H(J))
      GNRM=GNRM+ABS(G(J))
10    DY=DY+H(J)*G(J)
C
C        REPEAT SEARCH IN DIRECTION OF STEEPEST DESCENT IF DIRECTIONAL
C        DERIVATIVE APPEARS TO BE POSITIVE OR ZERO.
      IF (DY) 11,54,54
C
C        REPEAT SEARCH IN DIRECTION OF STEEPEST DESCENT IF DIRECTION
C        VECTOR H IS SMALL COMPARED TO GRADIENT VECTOR G.
11    IF (HNRM/GNRM-EPS) 54,54,12
C
C        SEARCH MINIMUM ALONG DIRECTION H
C
C        SEARCH ALONG H FOR POSITIVE DIRECTIONAL DERIVATIVE
12    FY=F
      ALFA=2.*(EST-F)/DY
      AMBDA=1.
C
C        USE ESTIMATE FOR STEPSIZE ONLY IF IT IS POSITIVE AND LESS THAN
C        1. OTHERWISE TAKE 1. AS STEPSIZE
      IF (ALFA) 15,15,13
13    IF (ALFA-AMBDA) 14,15,15
14    AMBDA=ALFA
15    ALFA=0.
C
C        SAVE FUNCTION AND DERIVATIVE VALUES FOR OLD ARGUMENT
16    FX=FY
      DX=DY
C
C        STEP ARGUMENT ALONG H
      DO 17 I=1,N
17    X(I)=X(I)+AMBDA*H(I)
C
C        COMPUTE FUNCTION VALUE AND GRADIENT FOR NEW ARGUMENT
      CALL FUNCT (N,X,F,G)
      FY=F
C
C        COMPUTE DIRECTIONAL DERIVATIVE DY FOR NEW ARGUMENT.  TERMINATE
C        SEARCH, IF DY IS POSITIVE. IF DY IS ZERO THE MINIMUM IS FOUND
      DY=0.
      DO 18 I=1,N
18    DY=DY+G(I)*H(I)
      IF (DY) 19,39,22
C
```

```
C        TERMINATE SEARCH ALSO IF THE FUNCTION VALUE INDICATES THAT
C        A MINIMUM HAS BEEN PASSED
19    IF (FY-FX) 20,22,22
C
C        REPEAT SEARCH AND DOUBLE STEPSIZE FOR FURTHER SEARCHES
20    AMBDA=AMBDA+ALFA
      ALFA=AMBDA
C        END OF SEARCH LOOP
C
C        TERMINATE IF THE CHANGE IN ARGUMENT GETS VERY LARGE
      IF (HNRM*AMBDA-1.E10) 16,16,21
C
C        LINEAR SEARCH TECHNIQUE INDICATES THAT NO MINIMUM EXISTS
21    IER=2
      RETURN
C
C        INTERPOLATE CUBICALLY IN THE INTERVAL DEFINED BY THE SEARCH
C        ABOVE AND COMPUTE THE ARGUMENT X FOR WHICH THE INTERPOLATION
C        POLYNOMIAL IS MINIMIZED
22    T=0.
23    IF (AMBDA) 24,39,24
24    Z=3.*(FX-FY)/AMBDA+DX+DY
      ALFA=AMAX1(ABS(Z),ABS(DX),ABS(DY))
      DALFA=Z/ALFA
      DALFA=DALFA*DALFA-DX/ALFA*DY/ALFA
      IF (DALFA) 54,25,25
25    W=ALFA*SQRT(DALFA)
      ALFA=DY-DX+W+W
      IF (ALFA) 26,27,26
26    ALFA=(DY-Z+W)/ALFA
      GO TO 28
27    ALFA=(Z+DY-W)/(Z+DZ+Z+DY)
28    ALFA=ALFA*AMBDA
      DO 29 I=1,N
29    X(I)=X(I)+(T-ALFA)*H(I)
C
C        TERMINATE, IF THE VALUE OF THE ACTUAL FUNCTION AT X IS LESS
C        THAN THE FUNCTION VALUES AT THE INTERVAL ENDS. OTHERWISE REDUCE
C        THE INTERVAL BY CHOOSING ONE END-POINT EQUAL TO X AND REPEAT
C        THE INTERPOLATION.  WHICH END-POINT IS CHOOSEN DEPENDS ON THE
C        VALUE OF THE FUNCTION AND ITS GRADIENT AT X
C
      CALL FUNCT (N,X,F,G)
      IF (F-FX) 30,30,31
30    IF (F-FY) 39,39,31
31    DALFA=0.
      DO 32 I=1,N
32    DALFA=DALFA+G(I)*H(I)
      IF (DALFA) 33,36,36
33    IF (F-FX) 35,34,36
34    IF (DX-DALFA) 35,39,35
35    FX=F
      DX=DALFA
      T=ALFA
      AMBDA=ALFA
      GO TO 23
36    IF (FY-F) 38,37,38
37    IF (DY-DALFA) 38,39,38
38    FY=F
      DY=DALFA
      AMBDA=AMBDA-ALFA
      GO TO 22
```

```
C
C     TERMINATE, IF FUNCTION HAS NOT DECREASED DURING LAST ITERATION
   39 IF (OLDF-F+EPS) 54,40,40
C
C     COMPUTE DIFFERENCE VECTORS OF ARGUMENT AND GRADIENT FROM
C        TWO CONSECUTIVE ITERATIONS
   40 DO 41 J=1,N
      K=N+J
      H(K)=G(J)-H(K)
      K=N+K
   41 H(K)=X(J)-H(K)
C        TEST LENGTH OF ARGUMENT DIFFERENCE VECTOR AND DIRECTION VECTOR
C        IF AT LEAST N ITERATIONS HAVE BEEN EXECUTED. TERMINATE, IF
C        BOTH ARE LESS THAN  EPS
      IER=0
      IF (KOUNT-N) 45,42,42
   42 T=0.
      Z=0.
      DO 43 J=1,N
      K=N+J
      W=H(K)
      K=K+N
      T=T+ABS(H(K))
   43 Z=Z+W*H(K)
      IF (HNRM-EPS) 44,44,45
   44 IF (T-EPS) 59,59,45
C
C        TERMINATE, IF NUMBER OF ITERATIONS WOULD EXCEED  LIMIT
   45 IF (KOUNT-LIMIT) 46,53,53
C
C        PREPARE UPDATING OF MATRIX H
   46 ALFA=0.
      DO 50 J=1,N
      K=J+N3
      W=0.
      DO 49 L=1,N
      KL=N+L
      W=W+H(KL)*H(K)
      IF (L-J) 47,48,48
   47 K=K+N-L
      GO TO 49
   48 K=K+1
   49 CONTINUE
      K=N+J
      ALFA=ALFA+W*H(K)
   50 H(J)=W
C
C        REPEAT SEARCH IN DIRECTION OF STEEPEST DESCENT IF RESULTS
C        ARE NOT SATISFACTORY
      IF (Z*ALFA) 51,1,51
C
C        UPDATE MATRIX H
   51 K=N31
      DO 52 L=1,N
      KL=N2+L
      DO 52 J=L,N
      NJ=N2+J
      H(K)=H(K)+H(KL)*H(NJ)/Z-H(L)*H(J)/ALFA
   52 K=K+1
      GO TO 5
C        END OF ITERATION LOOP
C
```

```
C        NO CONVERGENCE AFTER  LIMIT  ITERATIONS
53    IER=1
      RETURN
C
C        RESTORE OLD VALUES OF FUNCTION AND ARGUMENTS
54    DO 55 J=1,N
      K=N2+J
55    X(J)=H(K)
      CALL FUNCT (N,X,F,G)
C
C        REPEAT SEARCH IN DIRECTION OF STEEPEST DESCENT IF DERIVATIVE
C        FAILS TO BE SUFFICIENTLY SMALL
      IF (GNRM-EPS) 58,58,56
C
C        TEST FOR REPEATED FAILURE OF ITERATION
56    IF (IER) 59,57,57
57    IER=-1
      GO TO 1
58    IER=0
59    RETURN
      END
```

APPENDIX C

EXAMPLES OF PROGRAM OUTPUT

Example 1 - White and Woods Data of 1959

SOURCE	TEMPERATURE	RHO	RHO CALCULATED	DIFFERENCE
WHW059	10.000000	0.003200	0.001020	0.002180
WHW059	15.000000	0.017000	0.007719	0.009281
WHW059	20.000000	0.051000	0.031382	0.019618
WHW059	25.000000	0.120000	0.086670	0.033330
WHW059	30.000000	0.230000	0.183421	0.046579
WHW059	40.000000	0.540000	0.499982	0.040018
WHW059	50.000000	0.950000	0.936325	0.013675
WHW059	60.000000	1.430000	1.436732	-0.006732
WHW059	70.000000	1.960000	1.964969	-0.004969
WHW059	80.000000	2.500000	2.501859	-0.001859
WHW059	90.000000	3.030000	3.038230	-0.008230
WHW059	100.000000	3.550000	3.570093	-0.020093
WHW059	120.000000	4.600000	4.615721	-0.015721
WHW059	140.000000	5.600000	5.637964	-0.037964
WHW059	160.000000	6.650000	6.640883	0.009117
WHW059	180.000000	7.650000	7.628653	0.021347
WHW059	200.000000	8.600000	8.604656	-0.004656
WHW059	220.000000	9.600000	9.571464	0.028536
WHW059	250.000000	11.000000	11.008536	-0.008536

A	THETA	SUM OF (DIFFERENCES)**2
39.950176	217.537354	0.009226

Example 2 - Cox Data of 1943

SOURCE	TEMPERATURE	RHO	RHO CALCULATED	DIFFERENCE
MCOX43	77.330000	2.460000	2.464481	-0.004481
MCOX43	273.200001	12.410000	12.389841	0.020159
MCOX43	373.400002	17.180000	17.193873	-0.013873

A	THETA	SUM OF (DIFFERENCES)**2
39.512384	210.770609	0.000619

Example 3 - Cox Data of 1943 and White and Woods Data of 1959

SOURCE	TEMPERATURE	RHO	RHO CALCULATED	DIFFERENCE
WHW059	10.000000	0.003200	0.000929	0.002271
WHW059	15.000000	0.017000	0.007037	0.009963
WHW059	20.000000	0.051000	0.028769	0.022231
WHW059	25.000000	0.120000	0.080298	0.039702
WHW059	30.000000	0.230000	0.171999	0.058001
WHW059	40.000000	0.540000	0.478832	0.061168
WHW059	50.000000	0.950000	0.910253	0.039747
WHW059	60.000000	1.430000	1.411158	0.018842
WHW059	70.000000	1.960000	1.943838	0.016162
WHW059	80.000000	2.500000	2.487599	0.012401
WHW059	90.000000	3.030000	3.032195	-0.002195
WHW059	100.000000	3.550000	3.572969	-0.022969
WHW059	120.000000	4.600000	4.637077	-0.037077
WHW059	140.000000	5.600000	5.677648	-0.077648
WHW059	160.000000	6.650000	6.698330	-0.048330
WHW059	180.000000	7.650000	7.703241	-0.053241
WHW059	200.000000	8.600000	8.695818	-0.095818
WHW059	220.000000	9.600000	9.678715	-0.078715
WHW059	250.000000	11.000000	11.139177	-0.139177
MCOX43	77.330000	2.460000	2.342040	0.117960
MCOX43	273.200001	12.410000	12.259842	0.150158
MCOX43	373.400002	17.180000	17.045341	0.134659

A	THETA	SUM OF (DIFFERENCES)**2
41.621598	223.468170	0.114080

REFERENCES

1. Campbell, I. A.; Caplin, A. D.; and Rizzuto, C.: Momentum Nonconservation and the Low-Temperature Resistivity of Alloys. Phys. Rev. Letters, vol. 26, no. 5, Feb. 1, 1971, pp. 239-242.
2. Mills, D. L.: Temperature Dependence of the Contribution to the Transport Properties from Electron-Phonon Scattering in Dilute Alloys. Phys. Rev. Letters, vol. 26, no. 5, Feb. 1, 1971, pp. 242-245.
3. Rosenberg, Harold M.: Low Temperature Solid State Physics. Clarendon Press, Oxford, 1963, ch. 4.
4. Meaden, George T.: Electrical Resistance of Metals. Plenum Press, 1965, ch. 3.
5. Bloch, Felix: Über die Quantenmechanik der Elektronen in Kristallgittern. Zeit. f. Physik, vol. 52, Dec. 17, 1928, pp. 555-600.
6. Bloch, F.: Zur Suszeptibilität und Widerstandsänderung der Metalle in Magnetfeld. Zeit. f. Physik, vol. 53, Feb. 12, 1929, pp. 216-227.
7. Bloch, F.: Zum Elektrischen Widerstandsgesetz bei tiefen Temperaturen. Zeit. f. Physik, vol. 59, Jan. 2, 1930, pp. 208-214.
8. Grueneisen, E.: Dependence of the Electrical Resistance of Pure Metals on Temperature. Aun. d. Physik, vol. 16, no. 5, Mar. 1933, pp. 530-540.
9. Fletcher, R.; and Powell, M. J. D.: A Rapidly Convergent Descent Method for Minimization. Computer J., vol. 6, 1963/64, pp. 163-168.
10. Davidon, William C.: Variable Metric Method for Minimization. Rep. ANL-5990 (rev.), Argonne National Lab., Nov. 1959.
11. Cox, Martha: Thermal and Electrical Conductivities of Tungsten and Tantalum. Phys. Rev., vol. 64, nos. 7 and 8, Oct. 1 and 15, 1943, pp. 241-247.
12. White, G. K.; and Woods, S. B.: Electrical and Thermal Resistivity of the Transition Elements at Low Temperatures. Phil. Trans. Roy Soc. (London), Ser. A, vol. 251, no. 995, 1959, pp. 273-302.
13. Blackman, M.: On θ Values in the Resistance of Metals. Proc. Phys. Soc., Sec. A., vol. 64, pt. 8, no. 380, Aug. 1, 1951, pp. 681-683.

TABLE I. - COX DATA OF 1943

Temperature, K	Total electrical resistivity, ρ, μohm-cm
77.33	2.46
273.2	12.41
373.4	17.18

TABLE II. - WHITE AND WOODS DATA OF 1959

Temperature, K	Total electrical resistivity, ρ, μohm-cm
10	0.0032
15	.017
20	.051
25	.12
30	.23
40	.54
50	.95
60	1.43
70	1.96
80	2.50
90	3.03
100	3.55
120	4.60
140	5.60
160	6.65
180	7.65
200	8.60
220	9.6
250	11.0
273	12.1
295	13.1

TABLE III. - BLOCH-GRUENEISEN CONSTANTS AS FUNCTION OF MAXIMUM TEMPERATURE

Maximum temperature, K	Constant of metal, A, μohm-cm	Characteristic temperature of resistivity, θ_R, K
50	27.730	185.664
60	31.233	194.136
70	34.525	202.274
80	36.568	207.447
90	37.530	209.995
100	37.905	211.035
120	38.588	213.063
140	38.726	213.490
160	39.372	215.580
180	39.777	216.932
200	39.813	217.056
220	40.010	217.758
250	39.950	217.537



Figure 1. - Fletcher-Powell algorithm.



Figure 2. - Main program.



Figure 3. - Fit of White and Woods data.



Figure 4. - Fit of Cox data.



Figure 5. - Fit of Cox data of 1943 with White and Woods data of 1959.



Figure 6. - Constant of metal as function of highest temperature used.



Figure 7. - Characteristic temperature of resistivity as function of maximum temperature used.

NASA-Langley, 1971 —— 26 E-6242

"The aeronautical and space activities of the United States shall be conducted so as to contribute . . . to the expansion of human knowledge of phenomena in the atmosphere and space. The Administration shall provide for the widest practicable and appropriate dissemination of information concerning its activities and the results thereof."

—NATIONAL AERONAUTICS AND SPACE ACT OF 1958

NASA SCIENTIFIC AND TECHNICAL PUBLICATIONS

TECHNICAL REPORTS: Scientific and technical information considered important, complete, and a lasting contribution to existing knowledge.

TECHNICAL NOTES: Information less broad in scope but nevertheless of importance as a contribution to existing knowledge.

TECHNICAL MEMORANDUMS: Information receiving limited distribution because of preliminary data, security classification, or other reasons.

CONTRACTOR REPORTS: Scientific and technical information generated under a NASA contract or grant and considered an important contribution to existing knowledge.

TECHNICAL TRANSLATIONS: Information published in a foreign language considered to merit NASA distribution in English.

SPECIAL PUBLICATIONS: Information derived from or of value to NASA activities. Publications include conference proceedings, monographs, data compilations, handbooks, sourcebooks, and special bibliographies.

TECHNOLOGY UTILIZATION PUBLICATIONS: Information on technology used by NASA that may be of particular interest in commercial and other non-aerospace applications. Publications include Tech Briefs, Technology Utilization Reports and Technology Surveys.

Details on the availability of these publications may be obtained from:

SCIENTIFIC AND TECHNICAL INFORMATION OFFICE

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION
Washington, D.C. 20546